Blasius Boundary Layer

APPH 4200 Physics of Fluids
Columbia University

Similarity Equation

In[1]:=
```
eq = f[η] D[f[η], {η, 2}] / 2 + D[f[η], {η, 3}] == 0
```

Out[1]= $\frac{1}{2} f[\eta]\, f''[\eta] + f^{(3)}[\eta] == 0$

In[2]:=
```
ηBig = 25.0;
```

In[3]:=
```
bc = {f'[ηBig] == 1, f[0] == 0, f'[0] == 0}
```

Out[3]=
```
{f′[25.] == 1, f[0] == 0, f′[0] == 0}
```

Finding the Solution

In[4]:=
```
sol = NDSolve[{eq} ~Join~ bc, f, {η, 0, ηBig}]
```

Out[4]=
```
{{f → InterpolatingFunction[{{0., 25.}}, <>]}}
```

Graph the Solution

In[5]:=
```
fBlasius = f /. First[sol]
```

Out[5]=
```
InterpolatingFunction[{{0., 25.}}, <>]
```

In[6]:=
```
gBlasius[η_] = D[fBlasius[η], η]
```

Out[6]=
```
InterpolatingFunction[{{0., 25.}}, <>][η]
```

In[7]:=
```
Plot[fBlasius[η], {η, 0, ηBig}, PlotLabel → "f[η]"]
```

Out[7]=



In[8]:=
```
Plot[gBlasius[η], {η, 0, ηBig / 2},
 PlotLabel → "g[η]", PlotRange → All, AxesLabel → {"η", "u/U∞"}]
```

Out[8]=



Cross Velocity

Normalized to the $\sqrt{\text{Re}}$ …

In[9]:=
```
uy[η_] = (1 / 2) (η gBlasius[η] - fBlasius[η])
```

Out[9]=
$$\frac{1}{2}\left(-\text{InterpolatingFunction}[\{\{0., 25.\}\}, <>][\eta] + \eta\ \text{InterpolatingFunction}[\{\{0., 25.\}\}, <>][\eta]\right)$$

In[10]:=
```
Plot[uy[η], {η, 0, ηBig / 2}, PlotLabel → "Normalized uy[η]",
 PlotRange → All, AxesLabel → {"η", "(uy/U∞) √Re "}]
```

Out[10]=



Normalized Shear Stress

At the plate surface…

In[11]:=
```
D[gBlasius[η], η] /. η → 0
```

Out[11]=
```
0.332057
```

Summary

Blasius provided the mathematically rigorous "similiarity" solution to the boundary layer thickness for uniform flow along a fixed plate. The similiarity transformation creates an nonlinear ordinary differential equation that can be easily integrated.